

# Computergestuetzte Mathematik zur Analysis

## Lektion 2

### ▼ Unterschied Ausdruck (Expression) und Funktion (Function)

```
> r := (a * x^2 + b*x + c);
```

$$r := ax^2 + bx + c \quad (1.1)$$

```
> f := x -> sin(x*Pi);
```

$$f := x \rightarrow \sin(x\pi) \quad (1.2)$$

```
> f(1/2);
```

$$1 \quad (1.3)$$

```
> R := unapply(r, (x, a, b, c));
```

$$R := (x, a, b, c) \rightarrow ax^2 + bx + c \quad (1.4)$$

```
> R(0, 1, -1, 3);
```

$$3 \quad (1.5)$$

```
> int(r,x);
```

$$\frac{1}{3} ax^3 + \frac{1}{2} bx^2 + cx \quad (1.6)$$

```
> int(R(x,a,b,c),x);
```


$$\frac{1}{3} ax^3 + \frac{1}{2} bx^2 + cx \quad (1.7)$$

### ▼ Graphen von Funktionen


```
> ausdruck := sin(x);
```

$$\text{ausdruck} := \sin(x) \quad (2.1)$$


```
> plot(ausdruck,x=0..2*Pi);
```




```
> plot(f,0..2);
```



```
> plot(cos, 0 .. 2*Pi);
```




```
> plot([sin(x), cos(x)], x = 0 .. 2*Pi, color = [red, green],  
      thickness = 2);
```




```
> optionen := color = [red, green], thickness = 2;  
           optionen := color = [red, green], thickness = 2  
> plot([tan, cot], 0 .. 2*Pi, optionen);
```

(2.2)




Was geht hier schief?

```
> plot([tan, cot], 0 .. 2*Pi, -10 .. 10, optionen);
```



```
> plot([tan, cot], 0 .. 2*Pi, -10 .. 10, optionen, scont = true);
```



### 3D-Funktionsgraphen


```
> f := exp(-(x^2 + y^2));
```

$$f := e^{-x^2 - y^2}$$


(3.1)

```
> plot3d(f, x = -3 .. 3, y = -3 .. 3);
```





```
> plot3d(f, x = -3 .. 3, y = -3 .. 3, shading = zhue, axes =  
boxed, orientation = [70, 16]);
```




```
> g := sin(x*y);
```


```
g := sin(x y)
```

(3.2)


```
> plot3d(g, x = -2*Pi .. 2*Pi, y = -1 .. 1, axes="boxed");
```



```
> plot3d(g, x = -2*Pi .. 2*Pi, y = -1 .. 1, scaling=  
"constrained");
```



```
> plot3d(g/sqrt(x^2+y^2), x = -6..6, y = -6..6, view=-1..1,  
numpoints =1024,lightmodel=light1);
```




## ▼ Implizite Graphen

```
> with(plots);
```


```
[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d,
conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d,
densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d,
graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams,
intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot,
logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d,
polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, rootlocus,
semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata,
textplot, textplot3d, tubeplot]
```

```
> implicitplot((2*x^2+y^2-1)*(x^2+y^2-1/2),x=-1..1,y=-1..1,
coloring=[blue,green],filledregions=true,grid=[100,100]);
```


(4.1)



```
> implicitplot(4*x^2-y^2-1,x=-1..1,y=-2..2,scaling="constrained");
```




```
> implicitplot3d(x^3+y^3+z^3+1 = (x+y+z+1)^3, x=-2..2, y=-2..2,  
z=-2..2,numpoints=6025);
```




```
> implicitplot3d([x^3+y^3+z^3+1 = (x+y+z+1)^3, x^2+y^2+z^2=2],  
x=-2..2,y=-2..2,z=-2..2,numpoints=6000);
```





```
> spacecurve([sin(t),cos(t),t],t=0..4*Pi);
```



```
> knot := [-10*cos(t) -2*cos(5*t) + 15*sin(2*t), 15*cos(2*t) +  
10*sin(t)-2*sin(5*t),10*cos(3*t),t=0..2*Pi]:  
> spacecurve(knot,thickness=4,numpoints=600);
```




## ▼ Bewegte Bilder

```
> f:= exp( -(x-t)^2);
```

$$f := e^{-(x-t)^2}$$

(5.1)

```
> animate( plot, [exp(-(x-t)^2), x=-10..10,thickness=3],t=0..5,  
trace=5,frames=50);
```




```
> u:= exp(-(x-t)^2);
```

$$u := e^{-(x-t)^2}$$


(1)

```
> animate( plot, [u,x=-8..8,thickness=3],t=0..5,trace=5,frames=50);
```



```
> animate( plot3d, [A*(x^2+y^2),x=-3..3,y=-3..3],A=-2..2,style=patchcontour);
```

$$A = -2.$$



```
> animate( implicitplot, [x^2+A*y^2=1,x=-2..2,y=-2..2],A=-2..2);
```

$$A = -2.$$

